/*

Matthew Levendusky 741
Proposal title: Can We Reduce Affective Polarization in the Mass Public?

HYPOTHESES

Stated-Hyp 1: Priming American national identity will reduce affective polarization in the mass public.

	Test-Hyp 1: Participants in the national identity prime condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	Test-Hyp 2: Participants in the national identity prime condition will prefer less social distance to outpartisans than participants in the control condition.
	Test-Hyp 3: Participants in the national identity prime condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	Test-Hyp 4: Participants in the national identity prime condition will be more willing to discuss politics with outpartisans than participants in the control condition.

Stated-Hyp 2: Weakening partisan social identification will reduce affective polarization in the mass public.

	Test-Hyp 5: Participants in the weaknesses of inparty condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	Test-Hyp 6: Participants in the weaknesses of inparty condition will prefer less social distance to outpartisans than participants in the control condition.
	Test-Hyp 7: Participants in the weaknesses of inparty condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	Test-Hyp 8: Participants in the weaknesses of inparty condition will be more willing to discuss politics with outpartisans than participants in the control condition.

Stated-Hyp 3: Self-affirmation will reduce affective polarization in the mass public.

	Test-Hyp 9: Participants in the self-affirmation condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	Test-Hyp 10: Participants in the self-affirmation condition will prefer less social distance to outpartisans than participants in the control condition.
	Test-Hyp 11: Participants in the self-affirmation condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	Test-Hyp 12: Participants in the self-affirmation condition will be more willing to discuss politics with outpartisans than participants in the control condition.

********************************************************************************

NOTES: 
	
1. The feeling thermometer DV could be coded as difference score or just the
outparty FT.
	
*/

clear all
use Levendusky741, clear

********************************************************************************
* Exclude participants without party
	cap drop inparty
	gen inparty = .
	replace inparty = 0 if q1 == 1
	replace inparty = 0 if q2c == 1
	replace inparty = 1 if q1 == 2
	replace inparty = 1 if q2c == 2
	label define inpartyl 0 "Democrat" 1 "Republican" 
	label values inparty inpartyl  
	tab inparty, mis
	drop if inparty == .

* RECODING

* experiment condition
	tab xtess205, mis
	tab xtess205, nolabel
	replace xtess205 = 0 if xtess205 == 4
	label define xtess205l 0 "Control" 1 "National Identity" 2 "Weakness Inparty" 3 "Self Affirmation"
	label values xtess205 xtess205l  	
	tab xtess205, mis
	
* DV: Feeling Thermometer
	tab q5, mis
	tab q6, mis
	gen affpol_FT = abs(q5-q6)
	replace affpol_FT = . if q5 == -1
	replace affpol_FT = . if q6 == -1
	tab affpol_FT, mis

* DV: Social Distance
	tab q10, mis
	gen affpol_SD = q10
	replace affpol_SD = . if affpol_SD == -1
	tab affpol_SD, mis

* DV: Threat
	tab q7, mis
	gen affpol_threat = 6 - q7 
	replace affpol_threat = . if affpol_threat == 7
	tab affpol_threat, mis

* DV: Discussion
	tab q8, mis
	gen affpol_dis = .
	replace affpol_dis = 8 - q8 if inparty == 1
	replace affpol_dis = q8 if inparty == 0
	replace affpol_dis = . if q8 == -1
	tab affpol_dis, mis

********************************************************************************

* ANALYSIS

* Test-Hyp 1: Participants in the national identity prime condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	reg affpol_FT i.xtess205 if xtess205 == 1 | xtess205 == 0
	// no support for H1 (p = .928)
	tess 1.xtess205 -, init(Levendusky741) bonf(4)
	
* Test-Hyp 2: Participants in the national identity prime condition will prefer less social distance to outpartisans than participants in the control condition.
	reg affpol_SD i.xtess205 if xtess205 == 1 | xtess205 == 0
	// no support for H2 (p = .863)
	tess 1.xtess205 -, bonf(4)
	
* Test-Hyp 3: Participants in the national identity prime condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	reg affpol_threat i.xtess205 if xtess205 == 1 | xtess205 == 0
	// no support for H3 (p = .718)
	tess 1.xtess205 -, bonf(4)
	
* Test-Hyp 4: Participants in the national identity prime condition will be more willing to discuss politics with outpartisans than participants in the control condition.
	reg affpol_dis i.xtess205 if xtess205 == 1 | xtess205 == 0
	// no support for H4 (p = .096)
	tess 1.xtess205 +, bonf(4)
	
* Test-Hyp 5: Participants in the weaknesses of inparty condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	reg affpol_FT i.xtess205 if xtess205 == 2 | xtess205 == 0
	// no support for H5 (p = .415)
	tess 2.xtess205 -, bonf(4)
	
* Test-Hyp 6: Participants in the weaknesses of inparty condition will prefer less social distance to outpartisans than participants in the control condition.
	reg affpol_SD i.xtess205 if xtess205 == 2 | xtess205 == 0
	// no support for H6 (p = .521)
	tess 2.xtess205 -, bonf(4)
	
* Test-Hyp 7: Participants in the weaknesses of inparty condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	reg affpol_threat i.xtess205 if xtess205 == 2 | xtess205 == 0
	// no support for H7 (p = .283)
	tess 2.xtess205 -, bonf(4)
	
* Test-Hyp 8: Participants in the weaknesses of inparty condition will be more willing to discuss politics with outpartisans than participants in the control condition.
	reg affpol_dis i.xtess205 if xtess205 == 2 | xtess205 == 0
	// no support for H8 (p = .316)
	tess 2.xtess205 +, bonf(4)
	
* Test-Hyp 9: Participants in the self-affirmation condition will be less affectively polarized on a feeling thermometer than participants in the control condition.
	reg affpol_FT i.xtess205 if xtess205 == 3 | xtess205 == 0
	// no support for H9 (p = .188)
	tess 3.xtess205 -, bonf(4)
	
* Test-Hyp 10: Participants in the self-affirmation condition will prefer less social distance to outpartisans than participants in the control condition.
	reg affpol_SD i.xtess205 if xtess205 == 3 | xtess205 == 0
	// no support for H10 (p = .382)
	tess 3.xtess205 -, bonf(4)
	
* Test-Hyp 11: Participants in the self-affirmation condition will see the outparty's policies as less threatening to the nation than participants in the control condition.
	reg affpol_threat i.xtess205 if xtess205 == 3 | xtess205 == 0
	// no support for H11 (p = .662)
	tess 3.xtess205 -, bonf(4)
	
* Test-Hyp 12: Participants in the self-affirmation condition will be more willing to discuss politics with outpartisans than participants in the control condition.
	reg affpol_dis i.xtess205 if xtess205 == 3 | xtess205 == 0
	// no support for H12 (p = .492)
	tess 3.xtess205 +, bonf(4)
	